                               Building a SAS Data Set --- 3

LAG function – assign the value of a variable in the last observation 

Example 3.1

data  external;

   infile  ‘a:\dataset2\china#1.dat’;

   input   year  total  exports  imports;

   lag_exp = lag(exports);

   ch_exp = exports - lag_exp;

   lag_imp = lag(imports);

   ch_imp = imports - lag_imp;

   drop  total;

run;

proc print data=external;

   title 'Example 3.1';

run;

            Example 3.1         11:00 Thursday, August 10, 2000   1

   OBS    YEAR    EXPORTS    IMPORTS    LAG_EXP    CH_EXP    LAG_IMP    CH_IMP

     1    1955      1.41       1.73        .          .         .          .

     2    1956      1.65       1.56       1.41       0.24      1.73      -0.17

     3    1957      1.60       1.51       1.65      -0.05      1.56      -0.05

     4    1958      1.98       1.89       1.60       0.38      1.51       0.38

     5    1959      2.26       2.12       1.98       0.28      1.89       0.23

     32    1986     27.01      33.08      25.92       1.09     34.33      -1.25

     33    1987     34.71      33.40      27.01       7.70     33.08       0.32

     34    1988     40.64      39.85      34.71       5.93     33.40       6.45

     35    1989     43.44      39.14      40.64       2.80     39.85      -0.71

 DROP and KEEP -- keep only  some of the variables

Example 3.2

data  external;

   infile  ‘a:\dataset2\china#1.dat’;

   input   year  total  exports  imports;

   lag_exp = lag(exports);

   ch_exp = exports - lag_exp;

   lag_imp = lag(imports);

   ch_imp = imports - lag_imp;

   keep  year  exports  imports  ch_exp  ch_imp; 

run;

proc print data=external;

   title 'Example 3.2';

run;

  Example 3.2         11:00 Thursday, August 10, 2000   3

    OBS    YEAR    EXPORTS    IMPORTS    CH_EXP    CH_IMP

     1    1955      1.41       1.73        .         .

     2    1956      1.65       1.56       0.24     -0.17

     3    1957      1.60       1.51      -0.05     -0.05

     4    1958      1.98       1.89       0.38      0.38

     32    1986     27.01      33.08       1.09     -1.25

     33    1987     34.71      33.40       7.70      0.32

     34    1988     40.64      39.85       5.93      6.45

     35    1989     43.44      39.14       2.80     -0.71
Combining SAS data sets 

Example 3.3a (concatenating SAS data sets)

data  temp98;

  infile  'a:\dataset3\tmp98.dat';

  input  day & $5.  temp  weather: $6.;

run;

proc  print data=temp98;

   title 'Example 3.3a -- temp98';

data  temp99;

  infile  'a:\dataset3\tmp99.dat';

  input  day & $5.  temp;

run;

proc  print data=temp99;

   title 'Example 3.3a -- temp99';

data  combined;

   set  temp98  temp99;

run;

proc print data=combined;

   title 'Example 3.3a -- combined';

run;

 Example 3.3a -- combined   11:00 Thursday, August 10, 2000

            OBS     DAY     TEMP    WEATHER

             1     Jan 1     33     sun

             2     Jan 2     35     sun

             3     Jan 3     45     clouds

             4     Jan 5     28     snow

             5     Jan 1     25

             6     Jan 2     20

             7     Jan 3     35

             8     Jan 6     40

             9     Jan 8     39

Example 3.3b ( interleaving SAS data sets)

data  temp98;

  infile  'a:\dataset3\tmp98.dat';

  input  day & $5.  temp  weather: $6.;

run;

proc sort  data=temp98;

   by day;

data  temp99;

  infile  'a:\dataset3\tmp99.dat';

  input  day & $5.  temp;

run;

proc sort  data=temp99;

   by day;

data  combined;

   set  temp98  temp99;

   by day;

run;

proc print data=combined;

   title 'Example 3.3b';

run;

              Example 3.3b        11:00 Thursday, August 10, 2000  24

                                 OBS     DAY     TEMP    WEATHER

                                  1     Jan 1     33     sun

                                  2     Jan 1     25

                                  3     Jan 2     35     sun

                                  4     Jan 2     20

                                  5     Jan 3     45     clouds

                                  6     Jan 3     35

                                  7     Jan 5     28     snow

                                  8     Jan 6     40

                                  9     Jan 8     39

Example 2.3c (one-to-one merging of SAS datas)

data  temp98;

  infile  'a:\dataset3\tmp98.dat';

  input  day & $5.  temp  weather: $6.;

run;

data  temp99;

  infile  'a:\dataset3\tmp99.dat';

  input  date & $5.  temp;

run;

data  combined;

   merge  temp98  temp99;

run;

proc print data=combined;

   title 'Example 3.3c';

run;
     Example 3.3c        11:00 Thursday, August 10, 2000 

        OBS     DAY     TEMP    WEATHER    DATE

         1     Jan 1     25     sun        Jan 1

         2     Jan 2     20     sun        Jan 2

         3     Jan 3     35     clouds     Jan 3

         4     Jan 5     40     snow       Jan 6

         5               39                Jan 8

Example 3.3dA (matched merging of SAS data sets)

data  temp98;

  infile  'a:\dataset3\tmp98.dat';

  input  day & $5.  tmp98  weather: $6.;

run;

data  temp99;

  infile  'a:\dataset3\tmp99.dat';

  input  day & $5.  tmp99;

run;

data  combined;

   merge  temp98  temp99;

     by day;

run;

proc print data=combined;

   title 'Example 3.3dA';

run;
  Example 3.3dA        11:00 Thursday, August 10, 2000

          OBS     DAY     TMP98    WEATHER    TMP99

           1     Jan 1      33     sun          25

           2     Jan 2      35     sun          20

           3     Jan 3      45     clouds       35

           4     Jan 5      28     snow          .

           5     Jan 6       .                  40

           6     Jan 8       .                  39
Example 3.3dB (matched merging of SAS data sets)

data  data1;

  infile  'a:\dataset3\data1.dat';

  input  color $  Nitems;

run;

proc  sort  data=data1;

   by color;

data  data2;

  infile  'a:\dataset3\data2.dat';

  input  color $  cost $ ;

run;

proc  sort  data=data1;

   by color;

data  combined;

   merge  data1   data2;

     by color;

run;

proc print data=combined;

   title 'Example 3.3dB';

run;

  Example 3.3dB        11:00 Thursday, August 10, 2000 

           OBS    COLOR    NITEMS     COST

            1     blue       10      $14.00

            2     blue       23      $14.00

            3     blue       15      $14.00

            4     red        11      $20.50

            5     red        19      $20.50

            6     red        24      $20.50

Add comment lines

Example 3.4 

*  build SAS date set 'data1';

data  data1;

  infile  'a:\dataset3\data1.dat';

  input  color $  Nitems;

run;

proc  sort  data=data1;

   by color;

data  data2;  * build SAS data set 'data2';

  infile  'a:\dataset3\data2.dat';

  input  color $  cost $ ;

run;

proc  sort  data=data2;

   by color;

/*  merge data1 and data2 to build data set 'combined' */;

data  combined;

   merge  data1   data2;

     by color;

run;

proc print data=combined;  /* print data set 'combined' */;

   title 'Example 3.4';

run;

The output is the same as the last program.
